



プログラミング演習 (10)

関数

中村, 橋本, 小松, 渡辺

目標



- Processing で関数に挑戦！
 - 機能をどんどん作ってみよう！
 - 円とか四角形だけじゃなくて、色々な図形描画を関数にしてみよう！
 - 判定も関数で！

関数



- 背景を塗りつぶす: `background(色);`
- 円を描く: `ellipse(x座標, y座標, 縦直径, 横直径);`
- 線を描く: `line(x1, y1, x2, y2);`
- 四角形を描く: `rect(x座標, y座標, 横幅, 縦幅);`
- 距離を計算する: `dist(x1, y1, x2, y2);`

などなど, これまでに色々な機能をもつ関数を利用してきた.

こういった関数を作ることが今回の目的

関数の定義



関数の返り値の型 関数名 (引数リスト)

```
{  
    関数内のいろいろな処理  
    (返り値がある場合は) return 返り値;  
}
```

返り値がない場合は void

x, y に半径 r の円を描く関数

```
void circle( int x, int y, int r )  
{  
    ellipse( x, y, r*2, r*2 );  
}
```

半径 r の円の面積を求める関数

```
float menseki( float r )  
{  
    return r*r*3.14;  
}
```

関数について



- 最初に登場した「`void setup(){ ... }`」や、「`void draw(){ ... }`」は、`setup`や`draw`には返り値が無いという事を意味している
 - ただ準備, ただ描画をするだけなので！

関数のポイント




- 関数の引数リスト(どういう値を受け取るか)に入力とする変数を用意し, その変数のみを利用して結果を返すのがポイント
 - 描画であれば入力された x, y からの相対座標
 - 計算であれば入力された変数同士の計算
- 関数の利用用途
 - 色々な描画機能
 - 色々な計算機能
 - 色々な判定機能

自分専用の便利関数をたくさん作ろう!

棒人間を描く



(Q) x, y 座標を指定すると棒人間を描いてくれる関数を作成せよ！




棒人間を描く



- 考え方

- 棒人間は, 顔の中心の座標 (x, y) を与えると, 勝手に体と手と足を描くものにする
- 棒人間の中心の座標を (x, y) としたときのそれぞれの座標を決める
- 返り値はなし (void)
 - 描画するだけ



棒人間を描く



- マウ斯卡ーソルの場所に棒人間を描く

```
void setup(){
    size( 400, 400 );
}
void drawHuman( int x, int y ){
    ellipse( x, y, 20, 20 );
    line( x, y+10, x, y+40 );
    line( x-20, y+10, x+20, y+10 );
    line( x, y+40, x-20, y+60 );
    line( x, y+40, x+20, y+60 );
}
void draw(){
    background( 255 );
    drawHuman( mouseX, mouseY );
}
```

予習問題





- マウスクリックされた場所に棒人間を描くプログラムを書いてみましょう
 - mousePressed で drawHuman
- 配列と繰り返しを使ってマウスカーソルに追尾する棒人間を100人描いてみましょう
 - 前回の資料を参考に！
- マウスカーソルに追尾する棒人間100人を, どんどん薄くしましょう(同上)
- 棒人間じゃない何かを描画する関数を作りましょう(ロボットでもOK！)

関数でアニメーション

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



(Q) マウスの位置に応じて黒目を動かすプログラムを作ってみよう！（目玉を1つの関数とする）



関数でアニメーション



- 考え方

- 目玉の場所 (x, y) と、黒目が向く方向を決めるための座標 (mx, my) を引数にする
- mx, my はマウス座標を入力
- 白目を (x, y) を中心とし適当な楕円で描く
- x と mx の位置関係で黒目の座標を決め黒目を描く
 - $mx < x - 10$ なら黒目を左へ (10は適当な値)
 - $mx > x + 10$ なら黒目を右へ
 - そうでなければ黒目を真ん中へ
- `drawEye` を2つ `draw()` の中に書く！
- `drawHuman` は前に作ったのを使う

関数でアニメーション



描画だけなので
返り値はなし

```
void setup(){  
    size( 400, 400 );  
}  
  
void drawEye( int x, int y, int mx, int my ){  
    strokeWeight( 3 );  
    fill( 255, 255, 255 );  
    ellipse( x, y, 60, 70 );  
    fill( 0, 0, 0 );  
    if( x > mx+10 ){  
        ellipse( x-10, y, 10, 10 );  
    } else if( x < mx-10 ){  
        ellipse( x+10, y, 10, 10 );  
    } else {  
        ellipse( x, y, 10, 10 );  
    }  
}  
  
void draw(){  
    background( 255 );  
    drawHuman( mouseX, mouseY );  
    drawEye( 160, 200, mouseX, mouseY );  
    drawEye( 240, 200, mouseX, mouseY );  
}
```

引数は4つ

x と mx の位置で
条件分岐

予習問題




- マウスのX座標だけを考慮して目を動かしているが，マウスが下にある時，上にある時で目を上下に動かしてみましよう（合計8方向）
 - （ヒント） y と my の関係を利用
- 棒人間を左から右にアニメーション（移動）し，その移動に応じて黒目の動きを変えましよう
 - （ヒント）棒人間の座標を引数にする！
- （挑戦）もう少し目の動きをスムーズにしてみましよう！

予習問題



- 繰り返しを利用して妖怪百眼を作ってみよう
– x, y の2つのループを用意する！




アナログ時計を描く

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



(Q) 400x400の画面上に現在の時間に合わせてアナログ時計を表示するプログラムを作るには？




アナログ時計を描く



• 考え方

- 現在の時間は `hour()` で、現在の分は `minute()` で、現在の秒は `second()` で取得可能
- 長針, 短針, 秒針は, 角度 a , 半径を r としたとき, 画面の中央 (x, y) から $(x+r*\cos(a), y+r*\sin(a))$ までの線分を描くことで表現可能 (a は $0 \sim 2\pi$ のラジアン)
- 角度 a をどうやって求めるかが問題



アナログ時計を描く



- 考え方(続き)

- 12時間で1周(360度), 同じく60分, 60秒で1周

- hour 時は, 角度として $\text{hour} * 30$

- minute 分は, 角度として $\text{minute} * 6$

- second 秒は, 角度として $\text{second} * 6$

- 90度Processingの角度は進んでいるので, 90度だけ角度をマイナスする

- $\text{kh} = \text{hour} * 30 - 90;$

- $\text{km} = \text{minute} * 6 - 90;$

- $\text{ks} = \text{second} * 6 - 90;$

- 長針, 短針, 秒針は適当に半径を決定

アナログ時計を描く [1]




```
void drawClock( int x, int y, int r ){
    ellipse( x, y, r*2, r*2 );
    float kh = hour()*60-90;
    float km = minute()*6-90;
    float ks = second()*6-90;
    float len_h = r*0.5;
    float len_m = r*0.7;
    float len_s = r*0.9;
    line( x, y, x+len_h*cos(radians(kh)), y+len_h*sin(radians(kh)) );
    line( x, y, x+len_m*cos(radians(km)), y+len_m*sin(radians(km)) );
    line( x, y, x+len_s*cos(radians(ks)), y+len_s*sin(radians(ks)) );
}
void draw(){
    background( 255 );
    drawClock( 200, 200, 190 );
}
```

予習問題



- マウスカーソルの場所に応じてアナログ時計を動かしてみましょ
- LondonとTokyoとNew Yorkの時計を表示するアナログ時計を横に並べてみましょう(時差考慮)
 - サマータイムで日本とロンドン間の時差が+8時間, 日本とNYの間の時差が-13時間



予習問題



- 現在のアナログ時計は短針が1時間に1回，分針が1分に1回しか動かないが，本当はもっと細かく動くはず．それをプログラムしよう！
- アナログ時計を改良し，もっとわかりやすく，豪華にしてみましよう
 - 色を変えたり，太さを変えたり
- 面白い時計を作ってみましよう
 - 自分専用の時計を作ろう！

実行ファイルを作る！



どのOS用のアプリを作る？

Export Application

Export!

出力された！

**Win なら exe を実行！
アプリが出来た！**

```
clock | Processing 1.5.1
File Edit Sketch Tools Help
New Ctrl+N
Open... Ctrl+O
Sketchbook
Examples...
Close Ctrl+W
Save Ctrl+S
Save As... Ctrl+Shift+S
Export Applet Ctrl+E
Export Application Ctrl+Shift+E
Preferences Ctrl+Command
Quit Ctrl+Q

noFill();
drawClock( 100, 100, 100, hour()+8, minute(), second());
drawClock( 300, 100, 100, hour(), minute(), second());
drawClock( 500, 100, 100, hour()-13, minute(), second());
fill( 0 );

int hour, int minute, int second )
/2;
PI/2;
PI/2;

then h:sin(kakudo_h) );
udo_m) );
udo_s) );

名前
application.linux
application.macosx
application.windows
clock.pde

名前
lib
source
clock.exe

clock
London Tokyo New York
22 KB
```

約数の数を求める関数



(Q) ある入力された数字の約数の数を求める関数をどう作るか？ また、その関数を使って数字と約数の数のペアを出力しよう

```
4  
17989: 2  
17990: 16  
17991: 6  
17992: 16  
17993: 4  
17994: 8  
17995: 8  
17996: 12  
17997: 8  
17998: 4  
17999: 4  
18000: 60
```

約数の数を求める関数



- 考え方
 - 引数は整数型の `num` にする
 - 約数を数える整数型の変数 `count` を用意
 - 整数型の変数 `i` (1から`num`まで1ずつ増やす) を用意し, `num` が `i` で割り切れたら `count` を1追加する
 - 最後に `count` の値を返す
 - `return count;`
 - `println` で数と, 返って来た値を表示する

約数の数を求める関数



```
int getNumberOfDivisor( int num ){  
    int i=1;  
    int count=0;  
    while( i<=num ){  
        if( num%i == 0 ){  
            count++;  
        }  
        i++;  
    }  
    return count;  
}
```

戻り値(この値が、呼び出し元に返る)

ここに戻る

```
void setup(){  
    for( int i=1; i<100000; i++ ){  
        println( i+": "+getNumberOfDivisor(i) );  
    }  
}
```

素数かどうかを判定



(Q) 関数の引数として入力した素数かどうかを判定する関数を作ろう. 素数なら1を, 素数じゃなければ0を返すようにしよう

- 引数は整数型の `num` にする
- 約数の数が2だったら素数!
- `count == 2` だったら1を, それ以外だったら0を返すようにしよう!
- 関数の戻り値が1だったら素数! と表示する

素数かどうかを判定



```
int isPrimeNumber( int num ){
```

```
    int i=1;
```

```
    int count=0;
```

```
    while( i<=num ){
```

```
        if( num%i == 0 ){
```

```
            count++;
```

```
        }
```

```
        i++;
```

```
    }
```

```
    if( count == 2 ){
```

```
        return 1;
```

```
    } else {
```

```
        return 0;
```

```
    }
```

```
}
```

```
void setup(){
```

```
    for( int i=1; i<100000; i++ ){
```

```
        if( isPrimeNumber( i ) == 1 ){
```

```
            println( i+"は素数です！" );
```

```
        }
```

```
    }
```

```
}
```

ちなみに、約数の数を数える
プログラムをそのまま使うだけでもOK

予習問題



- 入力された数字が完全数かどうかを判定する関数を作ってみましょう
- 四角形と円の当たり判定を行う関数を作ってみましょう
- 四角形同士の当たり判定を行う関数を作ってみましょう

予習問題



- 過去に作ったロボットを関数化しましょう！
 - ロボットの部位ごとに関数化し，マウスなどによってアニメーションできるようにしてみましょう
 - また，draw() 内をどんどんシンプルにしましょう
- 過去に作ったゲームを関数化しましょう！
 - ゲームの機能を関数に分離しましょう
 - スコアを表示する関数を作っても良いかも！